@echo off
goto endTitle
============================================================================

    Surveyor Oracle Compare

============================================================================
 Copyright 2011, Matt Peloquin and Construx. This file is part of Code
 Surveyor, covered under GNU GPL v3 and is distributed WITHOUT ANY WARRANTY
============================================================================
:endTitle

REM Compare oracle files to output for a log
fc TestOracle\* output\* > output\OracleTest_Output.txt

REM Now create a file that only has diffs of oracle
fc TestOracle\* output\* | FIND "****" > output\OracleTest_Errors.txt

REM Crazy DOS way to get size of file
for %%A in (output\OracleTest_Errors.txt) do SET ORACLE_FAIL_SIZE=%%~zA

if %ORACLE_FAIL_SIZE% EQU 0 goto SUCCESS

echo.
echo   ORACLE CHECK FAILED 
echo   Oracle Diff Size: %ORACLE_FAIL_SIZE%
echo.
echo   You unzipped TestFiles.zip and TestOracle.zip, right?
echo.
(SET ORACLE_COMPARE_GOOD=)
goto :END

:SUCCESS
echo.
echo   Oracle matches output
echo.
(SET ORACLE_COMPARE_GOOD=1)

:END
echo.



